

Inspect_2_2.praat

Task:

Script to display a single or many sound files (and associated TextGrid files, if they exist) as a whole or of specific intervals. The files can be typed in, listed in a text file, or are listed in result-file of a previously analysis (e.g. the output of a Formant, Pitch, Spectrum, or Intensity script). Similarly, interval or point labels can be typed in or listed in a text file. Associated TextGrid files must have the same name as the sound file (i.e., only their extensions differ). If TextGrid files exist, they are always updated by this script (see page 6), whether boundaries or points are changed or not. The script can be exited at any time and it will continue with those files and intervals/points that had not been inspected during a previous session. With a small modification in the script, additional judgments (like ‘voiced’, ‘voiceless’), notes (position of wind and cursor), and/or text comments can be stored in text files.

The script creates a file named *000_Inspect_progress_V03.txt* which will be deleted when all specified intervals or points are processed. This file is used in case the user had interrupted a session to continue later with a new session at the last displayed interval without asking for any parameters. Deleting this file will cause the script to begin by asking for parameters and not to continue an interrupted session.

Contents:

Intro (this page)	1
Example for using a result file	2
Example for handling all sound files in a directory	3
Example for using a list of sound files	4
Example for displaying a single sound file centered at a time point.....	5
Example of the display of a window in a session	6
Example of the display of a window in a session with notes, judgments and comments	7
Description of parameters	9
Behavior during the execution of the script	11
Programming: adapting to different result-file styles	12
Programming: defining judgments options	13
Programming: explanations of some script mechanisms	14
Programming: description of some internal parameters	15

Parameter:

The screenshot shows a dialog box titled "Pause: Inspect parameters:". It contains several input fields and a checkbox, each with a callout box explaining its function:

- Directory:** A text input field with a callout "Directory path".
- Files:** A text input field with a callout "File specification".
- Labels:** A text input field with a callout "Label or time-point specification".
- Tier:** A text input field with a callout "Tier for label selection".
- Window:** A text input field containing the value "50" with a callout "Time adjustment for displayed signal".
- Always use default parameter settings?:** A checkbox labeled "Default" with a callout "Reset Formant, Pitch, Spectrogram to default parameters".

At the bottom of the dialog are three buttons: "Undo", "Stop", and "Continue".

Example for using a result file:

A previous analysis has generated a result file, e.g. *formant_results_210514_182809.txt*:

File	Label	Start(s)	Duration(ms)	Pitch_mean(Hz)	F1_mean(Hz)	F2_mean(Hz)
g071a000	u:	1.0139	31.6	127.59	309.6	1104.6
g071a000	a	1.1676	60.3	134.61	552	1434.8
g071a000	@	1.3639	29.3	128.69	355.5	1562.6
g071a000	a	1.5871	53.9	144.85	442.2	1307.2
g071a000	I	1.7069	25.3	139.68	292.5	2158
g071a000	a	1.9855	28.8	134.35	496.3	1113.3
g071a000	a:	2.1306	88.6	117.62	733.9	1255.8
g071a000	a	2.2194	111.7	118.01	714.6	1366.5
g071a000	U	2.4297	111.4	176.69	360.2	1358.6
g071a000	O	2.7113	47.1	148.77	468.3	1068.3
g071a000	i:	2.8991	18.9	0	NA	NA
g071a000	a:	2.9907	57.9	143.93	478.2	1212.9

The user wants to examine all labels [a, a:] and wants to add 50 ms on each side of the intervals:

Pause: Inspect parameters:

Leave the directory path for .wav files empty if you want to use the current

Directory: Data is in the present directory

Sound files (list) or result file of analysis:

Files: Result file of the formant analysis

Label(s) or center time

Labels: Labels to investigate

Tier number in case you want to select labels from TextGrids:

Tier: Not used for result-file processing

(Additional) window size in ms:

Window: Add 50 ms to each interval

Always use default parameter settings?

☐ Default Use the Pitch, Formant, Spectrogram parameters as the user sets them, even when the next interval is displayed

Example for handling all sound files in a directory:

A user wants to inspect all intervals with the label 'suit' in tier 2 of all sound files in a directory and wants to add 0.1 second to both ends of the intervals. (S)he wants to make sure that subsequent displays use the default parameters in case (s)he changes parameter of the spectrogram when inspecting one interval:

The screenshot shows the 'Pause: Inspect parameters:' dialog box in Praat. The settings are as follows:

- Directory:** /Users/henning/TIMIT (Annotated: Sounds and TextGrids are in this directory)
- Sound files (list) or result file of analysis:**
 - Files:** (Empty text field, Annotated: Handle all files)
- Label(s) or center time:**
 - Labels:** suit (Annotated: Show all intervals with the label *suit*)
- Tier number in case you want to select labels from TextGrids:**
 - Tier:** 2 (Annotated: The word labels are in tier 2)
- (Additional) window size in ms:**
 - Window:** 100 (Annotated: Add 100 ms before and after each interval)
- Always use default parameter settings?**
 - ☒ Default (Annotated: Always use default Pitch, Formant, Spectrogram parameters for each new interval)

Buttons at the bottom: Undo, Stop, Continue.

Example for using a list of sound files:

A user wants to display a list of sound files, e.g. given in the file *file_list.txt*:

g071a000
g071a001
g071a002
g071a003
g071a004
g071a005
g071a006
g071a007
g071a008
g071a009
g071a010

Pause: Inspect parameters:

Leave the directory path for .wav files empty if you want to use the current

Directory:

Sound files (list) or result file of analysis:

Files:

Label(s) or center time

Labels:

Tier number in case you want to select labels from TextGrids:

Tier:

(Additional) window size in ms:

Window:

Always use default parameter settings?

☐ Default

Undo Stop Continue

Callout boxes:

- Directory: Data is in the present directory
- Files: List of sound file names
- Labels: Do not select labels, i.e. use whole file
- Tier: Not used (since no label selection)
- Window: Display the whole file
- Default: Do not reset Pitch, Formant, Spectrogram parameters changes

Example for displaying a single sound file centered at a time point:

The user wants to display a 500 ms window of the file g071a000 around 2.35 s:

The screenshot shows the 'Pause: Inspect parameters:' dialog box in Praat. The dialog has several input fields and a checkbox, with callout boxes explaining the values entered:

- Directory:** The input field is empty. Callout: "File is in the present directory".
- Sound files (list) or result file of analysis:**
 - Files:** The input field contains "g071a000". Callout: "Sound file name".
 - Label(s) or center time:**
 - Labels:** The input field contains "2.35". Callout: "Center display at 2.35 s".
- Tier number in case you want to select labels from TextGrids:**
 - Tier:** The input field is empty. Callout: "Not used (since no label selection)".
- (Additional) window size in ms:**
 - Window:** The input field contains "500". Callout: "Display a 500 ms window".
- Always use default parameter settings?**
 - The checkbox is unchecked. Callout: "Do not reset Pitch, Formant, Spectrogram parameters changes".

At the bottom of the dialog are three buttons: "Undo", "Stop", and "Continue".

Example of the display of a window in a session:

The user had requested to show all files with a segment [a:] in tier 4 and to show additional 50 ms before and after each segment. (S)he has already inspected several windows and went back two windows to re-check something:

This is the (internal) list of intervals that will be handled (see page 12 for details).

	\$n+	##d	\$-h	\$a- a:	\$f	
1						original (204/215)
2	n	d	-h	a- a:	f	segment (215)
3	+n	d	dh	a:	f	realized (215)
4	n	d	dh	a:	f	basic (215)

Visible part 0.11375 seconds
Total duration 12.732625 seconds

all in out sel bak

Pause: Do what ever you want

Exit Previous **Next new** Next in list

The display was centered in the middle of of the label a: and 50 ms were added to both sides of the window. The user is free now to do normal PRAAT operations.

Clicking on *Next in list* will move the window to the next a: in the list of labels to inspect, even if that has been inspected (this will update the TextGrid).

Clicking on *Next new* will move the window to the next a: that has not been inspected yet (this will update the TextGrid).

Clicking on *Previous* will move the window to the previous a: in the list of labels to inspect (this will update the TextGrid).

Clicking on *Exit* will leave the script (without updating the TextGrid). The script will start the next time it runs with the last non-inspected segment.

Example of the display of a window in a session with notes, judgments and comments: @@@


The user had requested to show all files with a segment [z] in tier 4 and to show additional 50 ms before and after each segment. The script will accept text comments in the ‘Pause’ window and a judgment can be placed whether the inspected interval is voiced or voiceless. The judgments will be stored on a line-by-line basis in a text file with the name

“001_Inspect_<date>_<time>_judgments.txt”, the comments are stored in

“001_Inspect_<date>_<time>_comments.txt”, and the notes are stored in

“001_Inspect_<date>_<time>_notes.txt”.

(To configure the judgments option see page 13.)

**Example of “001_Inspect_210707_170540_judgments.txt”:**

file	label	tier	start	judgment
g071a000	z	4	2.8408	Voiceless
g071a000	z	4	3.9541	Voiceless
g071a000	z	4	5.6077	Voiced
g071a000	z	4	5.9428	Voiced
g071a000	z	4	7.0841	Voiced

Example of “001_Inspect_210707_170540_comments.txt”:

(note that not all interval have a comment)

file	label	tier	start	comment
g071a000	z	4	7.0841	Plosion at the beginning
g071a000	z	4	11.6167	hard to decide
g093a010	z	4	1.4266	Partly voiced
g093a010	z	4	2.1157	Clear voicebar

Example of “001_Inspect_210602_170540_notes.txt”:

(note that not all interval have a note)

file	start	end	cursor
g071a000	7.0341	7.2162	7.1251
g071a000	11.5667	11.7142	11.6405
g093a010	1.3766	1.5799	1.4783

Note that these files themselves can be used as a ‘result-file’ for Inspect.praat!

Description of parameters:

Directory:

The script handles all sound and TextGrid files in a directory. The path of this directory can be specified in this field. If this field is left empty, the script will handle all sound files in the directory where the script was started (i.e., the script is placed in the same directory as the sound and TextGrid files). The user can specify different directories for sound, TextGrid, and result files by changing variables in the script (see page 13). (Search for “### 1 >>>” and “### 4 >>>” in the script for directory specifications.)

Files:

The script can handle different set of files depending on the input given in this field:

<empty>: All sound files in a directory will be handled.

<file_name>.txt: The action depends on the type of file:

- File has only one column: each line will be used as name for a sound file.
- File has a result-file header: information from result file will be taken (see page 11 for details).

<file_name>: A sound file with the name <file_name> will be handled.

Labels:

The user can specify labels of intervals and points that should be displayed. In case the user has specified a single file or a list of files, the field **Tier** must also be specified. In case a previous result-file is used, the result-file must have a column for labels and **Tier** must not be specified. Labels can be specified in several ways:

<empty>: The display depends on the specification of the **Files** and **Window** fields:

- Files specified and Windows empty: displays the whole file
- Files and Window specified: displays the first milliseconds as specified by Window
- Result-file specifies intervals: display all intervals in the result file and add the ms given in Window before and after the interval
- Result-file specifies points: display all points in the result file and use the ms given in Window as window size

<one or more labels separated by spaces>: Search for any of the labels in the specified sound and Textgrid files and display intervals with time added given by Window or points and use Window as the window size to display.

<label_list_file.txt>: Search for any of the labels listed on a line-by-line basis in a text file (which must end with .txt) in the specified sound and TextGrid files and display intervals with time added given by Window or points and use Window as the window size to display.

Tier:

When labels are specified and sound files are specified (either by leaving the **Files** field empty, specifying files there or giving a list of files) **Tier** must be specified (otherwise, the script does not know where to find the labels, or same labels might be on different tiers). In case a result-file is specified in **Files**, or no **labels** are defined, the **Tier** field will be ignored.

Window:

A time in milliseconds. If intervals are defined, **Window** milliseconds are added before and after each interval. If points are defined, **Window** defines the size of the displayed window around the point location. If whole files are to be displayed (i.e., **Labels** is left empty) **Window** defines the size of the displayed window; or the whole file is displayed in this case, if the **Window** field is left empty.

Always use default parameter settings:

During the inspection of files, the user might interactively change Formant, Spectrogram or Pitch parameters. If this flag is not set, the script will continue with the changed parameters. If this flag is set, for subsequent files these parameters are reset to


Spectrogram settings: 0, 5000, 0.005, 70

Pitch settings: 75, 500, "Hertz", "autocorrelation", "automatic"

Formant settings: 5000, 5, 0.025, 30, 1

Behavior during the execution of the script:


The script display the waveforms along with the selected windows (spectrogram, pitch, formants, intensity as selected by the user) and pauses with a ‘Pause’ window. After inspection of the file, which can include changing any Tier of the TextGrids the user can click on ‘Next’ in the ‘Pause’ window to inspect the next item as given by the list file. The script keeps tract of inspected items in the file *000_Inspect_progress_V02.txt* by changing the ‘state’ value from ‘0’ to ‘1’ for the displayed interval or point.

**Behavior when using the ‘Exit’ option:**

‘Exit’ will stop the script without updating the TextGrid of the displayed window. The user can continue in a later session and the script will continue with those items in the list file that have not been already inspected. (In case a user changed boundaries of labels in a Tier but does not want to save them, (s)he can use ‘Exit’ to avoid updating the TextGrid and start the session again – it will continue with the last displayed interval.)

Behavior when using the ‘Previous’ option:

‘Previous’ will go back to the previous item in the list file (and update the currently displayed TextGrid). If more than one ‘Previous’ is used (i.e., the user went back more than one item, e.g. from the 10th item to the 7th item) the user gets two ‘Next’ options:



Choosing the ‘Next in list’ will move to the next item in the list (the 8th item in the example). Choosing the ‘Next new’ will jump to the next untreated item (i.e. the 10th item in the example).

Cancelling a session:

As long as the script finds the file *000_Inspect_progress_V02.txt* it will continue with the first non-inspected item. In case a user does not want to continue with a session and inspect other files, (s)he can simply delete the file *000_Inspect_progress_V02.txt* (or rename it temporarily, in case (s)he wants to work on this session later again).

Programming: adapting to different result-file styles:

In case a result-file (i.e. output of a previously ran analysis in a tab-delimited text file) exists, the script looks for a one-line header with certain keywords. The script needs at least a sound file name and a time (start time of an interval or time point). For intervals, additionally a duration or an end time must be specified. Furthermore, a label (interval or point) can be present in the result-file listing. For labels, also the tier number must be specified.

The script maps the words in the header of a result-file to the internal (column-)variables with a mechanism described here. The script uses for its internal representation the strings *file_string*\$, *label_string*\$, *start_string*\$, *duration_string*\$, and *end_string*\$. The words in the result-file are mapped onto these by assigning a string (e.g., if the result-file has a header word *Recording* to specify a file, the assignment *file_string*\$ = "*Recording*" binds the column *Recording* of the result-file to the internal *file_string*\$). Additionally, it can be specified whether duration is given in seconds or milliseconds (start and end times are always seconds). By changing the assignments in the scripts, the mapping from the keywords of the header in the result-file can be changed (search for "### 3 >>>" in the script for this code):

```
file_string$ = "File"
label_string$ = "Label"
start_string$ = "Start(s)"
duration_string$ = "Duration(ms)"
end_string$ = ""
# duration in the result file is given in in milliseconds or seconds
#     duration_is_ms = 0  seconds
#     duration_is_ms = 1  milliseconds
duration_is_ms = 1
```

The script copies this mapping into an array *result_header_<i>\$* to be later able to search for these words in the result-file header:

```
# make a tab-delimited string of the header items
result_header_items$ = file_string$ + tab$ + label_string$ + tab$ + start_string$ + tab$
                        + duration_string$ + end_string$
# use this string to put the words on a line-by-line basis
Create Strings from tokens: "help", result_header_items$, tab$
max_result_header = Get number of strings
# now copy the words into an array that will be used later for searching the header items
for i_col to max_result_header
    result_header_'i_col'$ = Get string: i_col
endfor
```

Later in the script (when a result-file is actually used) this array is used to build a hash to point to the specific columns:

```
for i_col to max_result_header
    header_col = Get column index: result_header_'i_col'$
    if (header_col)
        hash[result_header_'i_col'$] = header_col
    else
        . . .
    endif
endfor
```

Eventually, the hash will point to the specific column in the result-file. E.g., if a key word *Recording* in column 3 of the header of the result-file will point to the file name, the script will use

```
hash[file_string$] = 3
```

to access the filename (because *result_header_1\$* will be *Recording* in this example).

Programming: defining judgment options:

The script can provide choice options in the “beginPause...endPause” block. To switch this option on, the variable “judgments_flag” must be set to “1” (see `### 2 >>>` in the script). The definition of the name for the judgment “Voicing” and the options for this choice (e.g. “voiced”, “voiceless”, “unsure”) can be specified with the variables “judgment_0\$”, “judgment_1\$”, “judgment_2\$”, and “judgment_3\$”. I.e., the pause window will display for this example:

```
Voicing: o voiced
         o voiceless
         o unsure
```

These variables, and the variable “nr_judgments” (for the number of judgments) are defined at the beginning of the script (search for `### 2 >>>` in the script):

```
judgment_0$ = "Voicing"
judgment_1$ = "Voiced"
judgment_2$ = "Voiceless"
judgment_3$ = "Unsure"
nr_judgments = 3
```

The code in the “beginPause...endPause” block looks like:

```
if (judgments_flag)
  choice: judgment_0$, 1
  for i to nr_judgments
    option: judgment_'i'$
  endfor
endif
```

The conversion of the chosen option into a string that is eventually stored in the file `001_Inspect_<date>_<time>_judgments.txt` is performed by the code (search for `### 2 >>>` in the script – the comments are here more extended):

```
if (judgments_flag)
# make sure the first letter of the string judgment_0$ is lower case
# and convert all non-alphanumeric symbols to underline
# because it will be used as a variable name
# In our example:
# judgment_0$ contains the string "Voicing"
# the first replace_regex puts the string "voicing" into variable_name$
# the second replace_regex does not change anything
variable_name$ = replace_regex$ (judgment_0$, "^.", "\\L&", 1)
variable_name$ = replace_regex$ (variable_name$, "\\W", "_", 0)

# get the value which is stored in the variable with the name variable_name$ -
# this is the number of the selected option
# In our example the next line would be interpreted by praat as
# selected_option = 'voicing'
# and if the second option was selected, the value in selected_option will be 2
selected_option = 'variable_name$'

# now store this information
# selected_option has the value 2 in our example, and
# judgment_'selected_option'$ will become judgment_2$, which was defined
# at the beginning of the script as "Voiceless"
appendFileLine: judgments_file$, base_name$, tab$, label$, tab$, tier, tab$,
                  start$, tab$, judgment_'selected_option'$
endif
```

Programming: explanations of some script mechanisms:

The script has essentially three parts:

- 1) Definition of some parameters (and filling result-file header keyword mapping, see page 10)
- 2) Inquiring information from the user in case a new session starts and converting this information into a *progress_table* or loading an exiting *progress_table* from the file *000_Inpsect_progress_V02.txt*.
- 3) Interactively going through the *progress_table*.

The *progress_table* stores only a flag whether a particular interval has been displayed or not, the name of the sound file (usually only without full path and without extension), start of this interval, the end of this interval, and a time where the cursor should be displayed. Note that ‘interval’ can be a whole file, the interval of a label (eventually with added time before and after), or the window around a point. Information about the original label (interval or point) is not present in the *progress_table* (labels will show up in the associated TextGrid when that stretch of time is displayed). All this information is generated during the second part of the script, when the information given by the user is evaluated and converted into each row of the table.

The first 7 rows of the *progress_table* are used to store some global parameters that the user specified in the second part of the script (or had defined in the script before calling it the first time for a session) or that are defined in the first section of the script:

state	file	start	end	cursor	
2	?	5	?	?	# first row of real data
3	?	0	?	?	# default switch
4	/Users/henning/CD/TIMIT/PRAAT/Male/	?	?	?	# sound directory
5	/Users/henning/CD/TIMIT/PRAAT/Male/	?	?	?	# TextGrid directory
6	?	4	?	?	# tier
7	./001_Inspect_210602_182406_comments.txt	?	?	?	# comments file
8	./001_Inspect_210602_182406_judgments.txt	?	?	?	# judgments file

to be continued...

Programming: description of some internal parameters:**Directories:**

The script uses internally separate strings for sound, TextGrid, result-file, and support files. Users who use separate directories for these directories can adjust these names in the script (or put them into PRAATs 'form' window).

```
support_directory$ = ""  
<directory$ defined by user interaction>  
sound_directory$ = directory$  
grid_directory$ = directory$  
result_directory$ = directory$
```

Sound file extension:

sound_ext\$: The default extension for sound files is ".wav".

Default parameters:

The default parameters can be changed at the appropriate place in the script and other default parameters can be added there.

to be continued...

Current version and date:

2.2, 7-jul-2021

Known problems:

This documentation needs polishing and extension.

The handling of grid_directory\$, result_directory\$, and support_directory\$ needs improvement if a complete path for the sound_directory is defined and TextGrids etc. are located in different directories.

Planned extension:

Search for labels in more than one tier (or, if only one tier exists, just use it)

Use the position of cursor for comments and judgments reporting, if the whole file is displayed.

Display judgment and comments info in case a 'notes' file is used as source.

Use a text file to load/store all parameters.

Contact:

henning.reetz@icloud.com